
   Quantitative Analysis
   labor-leisure trade-off

    The purpose of this notebook is to illustrate a few functions of a
   single-variable and to introduce you to differentiation, single-
   variable optimization and comparative statics.

   To make the task more interesting, I have set up a simple 
   labor-leisure trade-off problem. In the problem, extra effort (less 
   leisure) in the initial period increases the agent's probability of 
   higher consumption in the second period. 
   
   In practice, you can think of the agent as a salaried employee who
   is trying to get a promotion (i.e. higher income/consumption). 

    We'll assume that the agent has no more opportunities to increase 
   his consumption ("get a promotion") after the initial period, so 
   the labor-leisure trade-off only occurs in the initial period.

   We will also assume that the agent is "risk-averse." In other words,
   if given the opportunity, the agent would purchase an "insurance" 
   policy that would pay a benefit if he/she did not get the "promotion," 
   but would require him/her to forego some of the "raise" if he/she
   did get the promotion.
   
   But because the agent can increase his/her probability of getting the
   promotion by working harder (i.e. foregoing leisure), he/she can 
   engage in "self-protection" (Ehrlich and Becker, JPE, 1972).

    In this problem, we will solve for the utility maximizing level of
   leisure and use comparative statics to show how it changes in response
   to variables that are exogenous to the agent. 

    Let's start by noting that the agent faces a time constraint. Each 
   additional hour that he/she spends on leisure is one less hour that 
   he/she can be working towards a "promotion."



qn-01_labor-leisure.wxm 2 / 17

(%i1) print("")$
print("define: ",T,"=",W,"+",L)$ 
print(T, " == total time (e.g. 24 hours, 365 days)")$
print(W, " == time spent working")$
print(L, " == time spent at leisure")$
print("")$
print("rearranging terms:")$
print(W," = ",T,"-",L)$
print("")$

define: T = W + L

T  == total time (e.g. 24 hours, 365 days)

W  == time spent working

L  == time spent at leisure

rearranging terms:

W  = T - L

    Next, let's define the probability of increased consumption (i.e.
   "promotion"). We'll assume that the probability increases with 
   time spent working (and decreases with time spent at leisure).

(%i10) print("")$
print("Assume that the probability of increased consumption ",
      "(\"promotion\") is:")$
p(L,T) := (1/2)*(1 + (log(T-L)/log(T)) )$
print("")$
print("p(L)"," = ",p(L,T))$
print("")$
print("NB: in Maxima, ",log(x)," means \"natural logarithm of x.\"")$
print("")$

Assume that the probability of increased consumption 

("promotion") is:

p(L)  = 

log T - L

log T
+ 1

2

NB: in Maxima, log x  means "natural logarithm of x."

(%i18) print("")$
print("Plotting probability as a function of leisure (with T = 24):")$
wxplot2d(p(L,24),[L,0.01,23.99],[y,0,1],
    [xlabel,"Leisure Time"],[ylabel,"Probability"])$
print("")$
print("Notice that if the agent works all the time (i.e.  L = 0), ")$
print("the probability goes to one. And if the agent indulges in ")$
print("leisure all the time, the probability goes to zero.")$
print("")$

Plotting probability as a function of leisure (with T = 24):

plot2d: some values were clipped.
(%t20) 

Notice that if the agent works all the time (i.e.  L = 0), 

the probability goes to one. And if the agent indulges in 

leisure all the time, the probability goes to zero.

    I chose this probability function for two reasons. First, the function
   exhibits diminishing marginal returns to effort (foregone leisure).
   Second, by defining the probability in terms of a ratio between the
   the natural log of time worked and the natural log of total time, what
   we obtain in practice is the base-T logarithm of time worked.

(%i26) print("")$
print("The ratio of the natural log of time worked to the natural log")$
print("of total time equals the base-T logarithm of time worked.")$
print("")$
print((log(T-L)/log(T))," = ",log[T](T-L))$
print("")$

The ratio of the natural log of time worked to the natural log

of total time equals the base-T logarithm of time worked.

log T - L

log T
 = logT T-L

    By using the base-T logarithm, we get two nice results. First, if the
   agent works all the time (i.e.  L = 0), the probability goes to one.
   Second, the probability goes to zero as leisure increases.

    We can also introduce an environmental variable that is exogenous to 
   the agent. For example, high unemployment might exogenously reduce the 
   agent's probability of "promotion." Alternatively, there may be 
   government policies that exogenously affect the agent's probability.

(%i32) /*  kill and replace the old function */
kill(p)$

print("")$
print("define:  ",g," == exogenous environmental variable, where: ",
      0,"<",g,"<",1)$
print("")$
p(L,T,g) := (g/2)*(1 + (log(T-L)/log(T)) )$
print("p(L,g)"," = ",p(L,T,g))$
print("")$

define:  g  == exogenous environmental variable, where: 0 < g < 1

p(L,g)  = 

g
log T - L

log T
+ 1

2

(%i39) wxplot2d([p(L,24,1),p(L,24,0.5)],[L,0.01,23.99],[y,0,1],
    [xlabel,"Leisure Time"],[ylabel,"Probability"],
    [legend,"high g","low g"])$
print("")$

plot2d: some values were clipped.
plot2d: some values were clipped.

(%t39) 

    We'll assume that the agent's utility function is an increasing 
   function of consumption and leisure. The agent makes decisions
   under uncertainty (i.e. there's a risk that he/she will not get the 
   "promotion"), so we'll include both a measure of risk-aversion in
   the utility function.

(%i41) print("")$
print("define: ")$ 
print(C_hi, " == consumption if agent gets \"promotion\"")$
print(C_lo, " == consumption if agent does not get \"promotion\" ")$
print("")$
print("assume that the risk averse agent's utility function is:")$
print("U(L) = p*log(C_hi) + (1-p)*log(C_lo) + a*log(L)")$
print("")$

define: 

C_hi  == consumption if agent gets "promotion"

C_lo  == consumption if agent does not get "promotion" 

assume that the risk averse agent's utility function is:

U(L) = p*log(C_hi) + (1-p)*log(C_lo) + a*log(L)

    The agent maximizes utility, so let's plot utility as a function of 
   leisure. Let's also see how the optimal level of leisure changes in 
   response to a change in the exogenous environmental variable "g."
   This will be our first "comparative statics" exercise.

(%i49) U(L,g) := ''(p(L,T,g)*log(C_hi) + (1-p(L,T,g))*log(C_lo) + a*log(L))$

/* assume some values */
a:(1/4)  $  C_lo:1  $  C_hi:3  $  T:24  $

/* optimal leisure when g=1 and when g=0.5 */
opt_ghi : lbfgs(-U(L,1.0), [L], [1.0], 1e-4, [-1,0])$
opt_glo : lbfgs(-U(L,0.5), [L], [1.0], 1e-4, [-1,0])$
L_ghi : subst(opt_ghi[1],L) $
L_glo : subst(opt_glo[1],L) $



qn-01_labor-leisure.wxm 7 / 17

(%i58) print("")$
wxplot2d([U(L,1),U(L,0.5)],[L,0.01,23.99],[y,0,1.8],
    [xlabel,"Leisure Time"],[ylabel,"Utility"],
    [legend,"high g","low g"])$
print("")$

plot2d: some values were clipped.
plot2d: some values were clipped.

(%t59) 

(%i61) print("")$
print("Notice that the optimal amount of leisure time is lower ",
      "when g is higher. ")$
print("")$
print("When g = 0.5, optimal leisure = ",round(10*L_glo)/10.0)$
print("When g = 1.0, optimal leisure = ",round(10*L_ghi)/10.0)$
print("")$
print("Agent works MORE (less leisure) when probability of ",
      "\"promotion\" increases.")$
print("Recall that the probability is an increasing function of \"g\").")$
print("")$

/* kill assumed values */
kill(a,C_lo,C_hi,T)$

Notice that the optimal amount of leisure time is lower 

when g is higher. 

When g = 0.5, optimal leisure = 18.2

When g = 1.0, optimal leisure = 14.2

Agent works MORE (less leisure) when probability of 

"promotion" increases.

Recall that the probability is an increasing function of "g").

    For the next comparative statics exercise, let's see how optimal 
   leisure changes when C_hi increases. (Note: When C_hi increases,
   the agent gets a larger raise with his/her promotion).

(%i71) /*  kill and replace the old function */
kill(U)$

(%i72) U(L,C_hi):=''(p(L,T,g)*log(C_hi) + (1-p(L,T,g))*log(C_lo) + a*log(L))$

/* assume some values */
a:(1/4)  $  C_lo:1  $  T:24  $  g:1  $

/* optimal leisure when C_hi=3 and when C_hi=2 */
opt_chi : lbfgs(-U(L,3.0), [L], [1.0], 1e-4, [-1,0])$
opt_clo : lbfgs(-U(L,2.0), [L], [1.0], 1e-4, [-1,0])$
L_chi : subst(opt_chi[1],L) $
L_clo : subst(opt_clo[1],L) $

(%i81) print("")$
wxplot2d([U(L,3),U(L,2)],[L,0.01,23.99],[y,0,1.8],
    [xlabel,"Leisure Time"],[ylabel,"Utility"],
    [legend,"high C_hi","low C_hi"])$
print("")$

plot2d: some values were clipped.
plot2d: some values were clipped.

(%t82) 

(%i84) print("")$
print("Notice that the optimal amount of leisure time is higher ",
      "when C_hi is higher.")$
print("")$
print("When C_hi = 2.0, optimal leisure = ",round(10*L_clo)/10.0)$
print("When C_hi = 3.0, optimal leisure = ",round(10*L_chi)/10.0)$
print("")$
print("Agent works MORE (less leisure) when reward for",
      "\"promotion\" is higher.")$
print("(Recall that a higher \"C_hi\" implies a higher \"raise.\").")$
print("")$

/* kill assumed values */
kill(a,C_lo,T,g)$

Notice that the optimal amount of leisure time is higher 

when C_hi is higher.

When C_hi = 2.0, optimal leisure = 16.9

When C_hi = 3.0, optimal leisure = 14.2

Agent works MORE (less leisure) when reward for "promotion" is higher.

(Recall that a higher "C_hi" implies a higher "raise.").

    The exercises above show how the agent's optimal leisure changes in 
   response to changes in exogenous variables. These are comparative 
   statics exercises.

   The question we did not address is how the agent determines his/her 
   optimal leisure in the first place. Instead, we simply looked at the 
   graph and found the level of leisure that maximizes utility.

    Just as a firm maximizes profit by producing output up to the point 
   where the marginal benefit of producing more output just offsets the 
   marginal cost, so our agent maximizes utility by indulging in leisure 
   up to the point where the marginal benefit of leisure just offsets
   the marginal cost.



qn-01_labor-leisure.wxm 11 / 17

(%i94) print("")$
print("Once again, assume the risk averse agent's utility function is:")$
print("")$
print("U(L) = p*log(C_hi) + (1-p)*log(C_lo) + a*log(L)")$
print("")$
print("The first two terms represent the total cost of leisure (lower")$
print("probability of promotion). The third term represents the total")$ 
print("benefit of leisure (leisure itself).")$
print("")$

Once again, assume the risk averse agent's utility function is:

U(L) = p*log(C_hi) + (1-p)*log(C_lo) + a*log(L)

The first two terms represent the total cost of leisure (lower

probability of promotion). The third term represents the total

benefit of leisure (leisure itself).

    The utility maximum is (by definition) the largest difference between 
   total benefit and total cost.

(%i103) TB(L) :=  a*log(L) $
TC(L) :=  -1*( p(L,T,g)*log(C_hi) + (1-p(L,T,g))*log(C_lo) )$
print("")$
print("TB(L) = ",TB(L))$
print("TC(L) = ",TC(L))$
print("")$

TB(L) = a log L

TC(L) = -log C_lo 1-

g
log T - L

log T
+ 1

2
-

log C_hi g
log T - L

log T
+ 1

2



qn-01_labor-leisure.wxm 12 / 17

(%i109) /* assume some values */
a:(1/4)  $  C_lo:1  $  C_hi:3  $  T:24  $  g:1  $

print("")$
wxplot2d([TB(L),TC(L)],[L,0.01,23.99],[y,-1.25,1.0],
    [xlabel,"Leisure Time"],[ylabel,"TB and TC"],
    [legend,"TB of leisure","TC of leisure"])$
print("")$

/* kill assumed values */
kill(a,C_lo,C_hi,T,g)$

(%t115) 

    The largest difference between benefit and cost (i.e. the utility 
   maximum) is found at the level of leisure where marginal benefit 
   equals marginal cost.
   
   The marginal benefit is the derivative (slope) of total benefit.
   The marginal cost is the derivative (slope) of total cost.

(%i118) /*  kill and replace the old function */
kill(U)$
U(L) := ''(p(L,T,g)*log(C_hi) + (1-p(L,T,g))*log(C_lo) + a*log(L))$

/* differentiate to obtain MB and MC */
MB(L):=''( diff( a*log(L), L) )$
MC(L):=''(-1*diff( p(L,T,g)*log(C_hi) + (1-p(L,T,g))*log(C_lo), L))$

/* simplify */
MC(L) := ''( ratsimp( MC(L) ))$

print("")$
print("MB(L) = ",MB(L))$
print("MC(L) = ",MC(L))$
print("")$

MB(L) = 
a

L

MC(L) = -
log C_lo - log C_hi g

2 T - 2 L log T

(%i127) /* assume some values */
a:(1/4)  $  C_lo:1  $  C_hi:3  $  T:24  $  g:1  $

print("")$
wxplot2d([MB(L),MC(L)],[L,0.01,23.99],[y,0,0.10],
    [xlabel,"Leisure Time"],[ylabel,"MB and MC"],
    [legend,"MB of leisure","MC of leisure"])$
print("")$

/* kill assumed values */
kill(a,C_lo,C_hi,T,g)$

plot2d: some values were clipped.
plot2d: some values were clipped.
(%t133) 

    The largest difference between benefit and cost (i.e. the utility 
   maximum) is found at the level of leisure where marginal benefit 
   equals marginal cost.

    Returning to comparative statics, let's look at how Marginal Cost 
   responds to an increase in "g." 

(%i136) /* kill the old function */
kill(MC)$
MC(L,g) := ''( diff(-1*( p(L,T,g)*log(C_hi) 
               + (1-p(L,T,g))*log(C_lo)), L))$

/* assume some values */
a:(1/4)  $  C_lo:1  $  C_hi:3  $  T:24  $

print("")$
wxplot2d([MB(L),MC(L,1),MC(L,0.5)],[L,0.01,23.99],[y,0,0.10],
    [xlabel,"Leisure Time"],[ylabel,"MB and MC"],
    [legend,"MB of leisure","MC when g = 1.0","MC when g = 0.5"])$
print("")$
print("The MC curve shifts upward when \"g\" increases, thus")$
print("reducing optimal leisure (i.e. the agent works more).")$
print("")$

/* kill assumed values */
kill(a,C_lo,C_hi,T)$

plot2d: some values were clipped.
plot2d: some values were clipped.
plot2d: some values were clipped.
(%t143) 

The MC curve shifts upward when "g" increases, thus

reducing optimal leisure (i.e. the agent works more).



qn-01_labor-leisure.wxm 16 / 17

    Now let's look at how Marginal Cost responds to an increase in "C_hi." 

(%i149) /* kill the old function */
kill(MC)$
MC(L,C_hi) := ''(diff(-1*( p(L,T,g)*log(C_hi) 
                 + (1-p(L,T,g))*log(C_lo)), L))$
/* assume some values */
a:(1/4)  $  C_lo:1  $  T:24  $  g:1  $
print("")$
wxplot2d([MB(L),MC(L,2),MC(L,3)],[L,0.01,23.99],[y,0,0.10],
    [xlabel,"Leisure Time"],[ylabel,"MB and MC"],
    [legend,"MB of leisure","MC when C_hi = 2.0","MC when C_hi = 3.0"])$
print("")$
print("The MC curve shifts upward when \"C_hi\" increases, thus")$
print("reducing optimal leisure (i.e. the agent works more).")$
print("")$

/* kill assumed values */
kill(a,C_lo,T,g)$

plot2d: some values were clipped.
plot2d: some values were clipped.
plot2d: some values were clipped.
(%t156) 

The MC curve shifts upward when "C_hi" increases, thus

reducing optimal leisure (i.e. the agent works more).

    Finally, let's algebraically find optimal leisure and show how it
   shifts in response to a change in "g" and "C_hi".

(%i162) L_opt(g,C_hi) := ''(subst(solve(MB(L) =  MC(L,C_hi) , L)[1],L))$
print("")$
print("L_opt = ",L_opt(g,C_hi)," > 0")$
print("")$
print(("d L_opt"/"d g")," = ",diff( L_opt(g,C_hi), g)," < 0")$
print("")$
print(("d L_opt"/"d C_hi")," = ",diff( L_opt(g,C_hi), C_hi)," < 0")$
print("")$
print("These expressions are complicated, but our analysis is simple.")$
print("All we're interested in is the sign (positive or negative).")$
print("The negative derivatives indicate that \"L_opt\" falls in")$
print("response to an increase in \"g\" or \"C_hi.\".")$
print("")$

L_opt = 
2 a T log T

2 a log T + log C_hi - log C_lo g
> 0

d L_opt

d g
 = -

2 a log C_hi - log C_lo T log T

2 a log T + log C_hi - log C_lo g
2
< 0

d L_opt

d C_hi
 = -

2 a g T log T

C_hi 2 a log T + log C_hi - log C_lo g
2
< 0

These expressions are complicated, but our analysis is simple.

All we're interested in is the sign (positive or negative).

The negative derivatives indicate that "L_opt" falls in

response to an increase in "g" or "C_hi.".


